
@Description
LatchKeys 1.0 - remap incoming keys to other keys/channels 
                with selectable latch and inversion functionality
Each mapping uses three knobs:
  - incoming note
  - outgoing note
  - outgoing channel / Latch (or not) / Invert (or not)
To reset a mapping, turn the third knob all the way to the left.
 
Original notes will not be passed through when a mapping occurs.
All other events are forwarded.        
@End



@OnLoad
   // Constants
   off=0
   on=1
   latchOff=99
   latchOn=100
   init=-1
   
   inIdx=0
   outIdx=1
   chIdx=2
   
   
   if Unassigned maxKeys
   
      maxKeys=6
      FillArray inKey, init, maxKeys
      FillArray outKey, init, maxKeys
      FillArray outCh, init, maxKeys
      FillArray latched, init, maxKeys
      FillArray inverted, init, maxKeys
      FillArray state, init, maxKeys
      knobIndex[] = [0, 4, 8, 11, 15, 19]
      for i=0 to 21
          SetKnobValue i, 0
      endfor
   endif
   
   ShowLayout 1
   LabelKnobs {LatchKeys 1.0 - map and latch keys}
   for i=0 to 21
       LabelKnob i,{ }
   endfor
   Call @Repaint
@End


@OnMidiNoteOn
    processed=false
    for i=0 to maxKeys-1
       if MIDINote=inKey[i]
          processed=true
          if state[i]=off
             state[i]=on
             if inverted[i]
                SendMidiNoteOff outCh[i], outKey[i], 127
             else
                SendMidiNoteOn outCh[i], outKey[i], 127
             endif
           elseif state[i]=latchOff
             state[i]=latchOn
             if inverted[i]
                SendMidiNoteOff outCh[i], outKey[i], 127
             else
                SendMidiNoteOn outCh[i], outKey[i], 127
             endif
           elseif state[i]=latchOn
             state[i]=latchOff
             if inverted[i]
                SendMidiNoteOn outCh[i], outKey[i], 127
             else
                SendMidiNoteOff outCh[i], outKey[i], 127
             endif
           endif
        endif
    endfor
    if processed=false
        SendMidiNoteOn MIDIChannel, MIDINote, MIDIVelocity 
    endif
@End

@OnMidiNoteOff
    processed=false
    for i=0 to maxKeys-1
       if MIDINote=inKey[i]
          processed=true
          if state[i]=on
             state[i]=off
             if inverted[i]
                SendMidiNoteOn outCh[i], outKey[i], 127
             else
                SendMidiNoteOff outCh[i], outKey[i], 127
             endif
           endif
        endif
    endfor
    if processed=false
        SendMidiNoteOff MIDIChannel, MIDINote, MIDIVelocity 
    endif
@End


@OnMidiInput
    if MIDICommand<>0x80 AND MIDICommand<>0x90
    then
        SendMIDIThru
    endif
@End


@OnKnobChange
    lk=LastKnob
    v = Round(GetKnobValue lk)
    // SetKnobValue LastKnob, v
    for i=0 to maxKeys-1
       idx=knobIndex[i]
       if lk = idx+inIdx
          inKey[i]=v
       elseif lk = idx+outIdx
          outKey[i]=v
       elseif lk = idx+chIdx
          if v=0
             // reset
             inKey[i]=-1
             outKey[i]=-1
             outCh[i]=-1
             latched[i]=-1
             inverted[i]=-1
             // null out the knobs also
             SetKnobValue idx+inIdx, 0
             SetKnobValue idx+outIdx, 0
             SetKnobValue idx+chIdx, 0
          else
             //outCh[i]=Div v, 8
             v=v-1
             if v >= 64
                v=v-64
                inverted[i]=true
             else
                inverted[i]=false
             endif
               
             outCh[i]=RoundDown(v / 4)
             v=v - (4*outCh[i])
             if v >= 2
                latched[i]=true
                state[i]=latchOff
             else
                latched[i]=false
                state[i]=off
             endif
          endif // v=0
       endif // LastKnob
   endfor // i
                
   Call @Repaint

@End


@Repaint
    for i=0 to maxKeys-1
        idx=knobIndex[ i ]
        
        if inKey[i] < 0
           LabelKnob idx+inIdx, {In}, i+1
        else
           LabelKnob idx+inIdx, (NoteName inKey[i], true), {/}, inKey[i]
        endif
        if outKey[i] < 0
           LabelKnob idx+outIdx, {Out}, i+1
        else
           LabelKnob idx+outIdx, (NoteName outKey[i], true), {/}, outKey[i]
        endif
        if outCh[i] < 0
           LabelKnob idx+chIdx, {OutCh}
        else
           if inverted[i]
              if latched[i]
                  LabelKnob idx+chIdx, {ch}, outCh[i]+1, {LatInv}
              else
                  LabelKnob idx+chIdx, {ch}, outCh[i]+1, {Invert}
              endif
           else
              if latched[i]
                  LabelKnob idx+chIdx, {ch}, outCh[i]+1, {Latch}
              else
                  LabelKnob idx+chIdx, {ch}, outCh[i]+1
              endif
           endif
        endif
    endfor
@end

